

**BỘ GIÁO DỤC VÀ ĐÀO TẠO**

**TRƯỜNG ĐẠI HỌC MỎ - ĐỊA CHẤT**

**BÁO CÁO HỌC THUẬT**

**Lập các chương trình con tìm nghiệm  
xấp xỉ của bài toán biên trên Maple**

---

*TS. Hoàng Ngự Huân*

**Hà Nội, 6/2022**

**BỘ GIÁO DỤC VÀ ĐÀO TẠO  
TRƯỜNG ĐẠI HỌC MỎ - ĐỊA CHẤT**

**BÁO CÁO HỌC THUẬT**

**Lập các chương trình con trên nghiệm  
xấp xỉ của bài toán biên trên Maple**

---

**Xác nhận của bộ môn**

**Hà Nội, 6/2022**

## Xấp xỉ hàm số

Vấn đề mấu chốt trong việc giải số các phương trình vi phân nằm ở phương pháp xấp xỉ hàm số. Vì vậy trước khi nghiên cứu các phương pháp xấp xỉ để giải bài toán biên chứa phương trình vi phân, chúng ta sẽ tiến hành mô tả các phương pháp xấp xỉ hàm số.

Với những bài toán được mô hình bởi phương trình vi phân, nghiệm thường đi kèm với đòi hỏi thỏa mãn điều kiện biên cho trước. Ví thể, đầu tiên ta xem xét phương pháp xấp xỉ hàm trong đó: trên biên  $G$  của miền  $\Omega$  nhận những giá trị giống hàm  $\varphi$  cho trước.

Giả sử, ta tìm được một hàm  $\psi$  nhận giá trị trên biên  $G$  giống hàm  $\varphi$ :  $\varphi|_G = \psi|_G$  và hệ các hàm cơ sở

độc lập tuyến tính  $N_m$ ,  $m = 1, 2, 3 \dots$ . Các hàm cơ sở này nhận giá trị 0 trên biên  $G$ :  $N_m|_G = 0$

$$\varphi \approx \psi + \sum_{m=1}^M a_m N_m$$

$$R_\Omega = \varphi - \psi - \sum_{m=1}^M a_m N_m$$

$$\int_{\Omega} W_m R_\Omega \, d\Omega = 0$$

$$\text{Phương pháp Galerkin } \int_{\Omega} N_m R_\Omega \, d\Omega = 0$$

```
> Bubnov_Galerkin := proc(M, a) local m, res, R, eq, eqns, var;
  a := array(1..M) :
  eqns := { } : var := { } :
  R := phi(x) - Phi(x, M) :
  for m from 1 to M do
    eq[m] := int(R * N(x, m), x = 0..1) = 0 :
    eqns := eqns union {eq[m]} :
    var := var union {a[m]} :
  end do:
  res := solve(eqns, var); assign(res) :
end proc:
```

>

**Bài toán 1.** Sử dụng hệ các hàm cơ sở thích hợp (dạng đa thức) để xấp xỉ hàm số

$\phi = 1 + \sin\left(\frac{\pi \cdot x}{2}\right)$  trên đoạn  $0 \leq x \leq 1$ . Áp dụng phương pháp phương pháp Bubnov-Galerkin. Với kết quả xấp xỉ thu được, hãy khảo sát số sự hội tụ về hàm đã cho.

```
> phi := x -> 1 + sin(pi*x/2);
p := plot(phi(x), x=0..1, color = black,
legend = `given function`);

phi := x -> 1 + sin(1/2 * pi * x) (1.1.1)
```

```
> psi := x -> x + 1; psi(0); psi(1);

psi := x -> 1 + x
1
2 (1.1.2)
```


```
> N := (x, m) -> x^m * (1-x);
Phi := proc(x, M) option operator, arrow; local m;
psi(x) + sum(a[m] * N(x, m), m = 1..M); end proc;
Phi(0, M); Phi(1, M);

N := (x, m) -> x^m * (1-x)
Phi := (x, M) -> psi(x) + sum_{m=1}^M a_m * N(x, m)
1
2 (1.1.3)
```

### ▼ Xấp xỉ một thành phần

```
M := 1; with(plots) :
Bubnov_Galerkin(M, a) :
Phi(x, M);
G1 := plot(Phi(x, M), x=0..1, color = black, style = point,
legend = `1 elementary approximation`);
plots[display]({G1, p});
```

$$1 + x - \frac{5}{2} \frac{(-192 + \pi^3 + 48\pi)x(1-x)}{\pi^3}$$



▼ **Xấp xỉ hai thành phần**

$M := 2;$

$Bubnov\_Galerkin(M, a) :$


$\Phi(x, M);$

$G2 := plot(\Phi(x, M), x=0..1, color=black, style=point,$

$symbol=cross, legend='2 elementary approximation');$

$plots[display]({G1, G2, p});$

$$1 + x + \frac{(10560 \pi + \pi^4 + 720 \pi^2 - 40320) x (1 - x)}{\pi^4} - \frac{7 (2880 \pi + \pi^4 + 240 \pi^2 - 11520) x^2 (1 - x)}{\pi^4}$$



## ▼ Xấp xỉ nghiệm của phương trình vi phân

Xét trong miền  $\Omega$  phương trình vi phân viết dưới dạng sau

$$A(\phi) = L\phi + p = 0 \quad (2.9)$$

Trong đó  $L$  là toán tử vi phân tuyến tính,  $p$  không phụ thuộc hàm  $\phi$

Nghiệm của phương trình (3.9) phải thỏa mãn các điều kiện biên trên biên  $\Gamma$  của miền  $\Omega$ . Tức là

$$B(\phi) = M\phi + r = 0 \quad (2.10)$$

Trong đó  $M$  là toán tử vi phân tuyến tính,  $r$  không phụ thuộc vào hàm  $\phi$ .

Theo phương pháp được trình bày ở trên, ta sẽ tiến hành xấp xỉ nghiệm  $\phi$  và sử dụng khai triển (3.1)

$$\phi \approx \psi + \sum_{m=1}^M a_m N_m$$

$$\phi \approx \left( \psi + \sum_{m=1}^M a_m N_m \right) \quad (2.1)$$

## Cách 1: các hàm $N$ trên biên bằng không, còn hàm $\psi$ thỏa mãn điều kiện biên.

**Bài toán 2.** Trong lý thuyết lực xoắn đàn hồi của thanh tròn hình lăng trụ, xuất hiện phương trình

$$\varphi_{xx} + \varphi_{yy} = -2.$$

Hàm  $\varphi$  bằng không trên biên  $-3 \leq x \leq 3, -2 \leq y \leq 2$ .

Lời giải. Vì nghiệm có tính chất đối xứng qua các đường thẳng  $x=0, y=0$  vì vậy ta chọn các hàm chẵn có các tính chất này

$$N_1 = \cos\left(\frac{\pi x}{6}\right) \cos\left(\frac{\pi y}{4}\right), N_2 = \cos\left(\frac{3\pi x}{6}\right) \cos\left(\frac{\pi y}{4}\right), N_3 = \cos\left(\frac{\pi x}{6}\right) \cos\left(\frac{3\pi y}{4}\right)$$

```
> Bas_func := proc(M, N) local m;  
  for m from 1 to M do  
    N[2 * m - 1] := cos(pi * x / 6) * cos((2 * m - 1) * pi * y / 4);  
    N[2 * m] := cos((2 * m + 1) * pi * x / 6) * cos(pi * y / 4);  
  end do;  
end proc;
```

Xác định hàm xấp xỉ bằng chương trình con

```
> psi := proc(M) option operator, arrow; local m;  
  sum(a[m] * N[m], 'm' = 1 .. M); end proc;
```

chương trình con lập các phương trình của phương pháp sai lệch đối trọng dưới dạng phương pháp Bubnov-Galerkin

```
> Bubnov_Galerkin := proc(M) local eq, m, R, eqns, var, res;  
  R := diff(psi(M), x$2) + diff(psi(M), y$2) + 2 :  
  eqns := { } : var := { } :  
  for m from 1 to M do  
    eq[m] := int(int(R * N[m], x = -3 .. 3), y = -2 .. 2) = 0 :  
    eqns := eqns union {eq[m]} :  
    var := var union {a[m]} :  
  end do;  
  res := solve(eqns, var) : assign(res) :  
  print(res);  
end proc;
```

Bây giờ ta chuyển qua lập nghiệp xấp xỉ của bài toán. Đầu tiên là tính một vài hàm cơ sở

```
> Bas_func(3, N) : eval(N);  
table([ [1 = cos(1/6 * pi * x) * cos(1/4 * pi * y), 2 = cos(1/2 * pi * x) * cos(1/4 * pi * y), 3
```

(2.1.1)

$$= \cos\left(\frac{1}{6} \pi x\right) \cos\left(\frac{3}{4} \pi y\right), 5 = \cos\left(\frac{1}{6} \pi x\right) \cos\left(\frac{5}{4} \pi y\right), 4$$

$$= \cos\left(\frac{5}{6} \pi x\right) \cos\left(\frac{1}{4} \pi y\right), 6 = \cos\left(\frac{7}{6} \pi x\right) \cos\left(\frac{1}{4} \pi y\right) \Big]$$

Ta thu được nghiệm với ba thành phần xấp xỉ

>  $M := 3; a := \text{array}(1..M);$   
 $\Psi(M); \text{Bubnov\_Galerkin}(M);$

$$M := 3$$

$$a := \text{array}(1..3, [ ])$$

$$a_1 \cos\left(\frac{1}{6} \pi x\right) \cos\left(\frac{1}{4} \pi y\right) + a_2 \cos\left(\frac{1}{2} \pi x\right) \cos\left(\frac{1}{4} \pi y\right)$$

$$+ a_3 \cos\left(\frac{1}{6} \pi x\right) \cos\left(\frac{3}{4} \pi y\right)$$

$$\left\{ a_1 = \frac{4608}{13 \pi^4}, a_2 = -\frac{512}{15 \pi^4}, a_3 = -\frac{1536}{85 \pi^4} \right\}$$

(2.1.2)

## ▼ Cách 2: Xấp xỉ nghiệm của phương trình vi phân và điều kiện biên

Ở phần trước, chúng ta đã trình bày cách giải xấp xỉ phương trình vi phân bằng cách khai triển theo các hàm cơ sở và xây dựng hàm xấp xỉ thỏa mãn mọi điều kiện biên của bài toán. Ở mục này, ta sẽ hạ thấp yêu cầu đối với hàm xấp xỉ; đồng thời với nó là mở rộng cách chọn dạng của các hàm cơ sở.

Chúng ta sẽ coi rằng khai triển

$$\varphi \approx \sum_{m=1}^M a_m N_m$$

$$\varphi \approx \left( \sum_{m=1}^M a_m N_m \right) \quad (2.2.1)$$

Không thỏa mãn điều kiện biên nào đó hoặc không thỏa mãn tất cả các điều kiện biên của bài toán ngay từ đầu. Khi đó, cùng với sai số theo miền

$$R_\Omega = A(\varphi) = L\varphi + p$$

xuất hiện thêm sai số ở các điều kiện biên

$$R_\Gamma = B(\varphi) = M\varphi + r$$

Bây giờ ta có thể làm giảm tổng số các sai số trên biên và trên miền bằng cách ép

$$\int_{\Omega} W1_{,n} R_\Omega \, d\Omega + \int_{\Gamma} W2_{,n} R_\Gamma \, d\Gamma = 0$$

Trong đó, một cách tổng quát, các hàm đối trọng  $W1$  và  $W2$  có thể chọn độc lập lẫn nhau

Để minh họa, ta lại xem xét bài toán 2 nhưng sẽ khai triển theo các hàm cơ sở không thỏa mãn các điều kiện biên ngay từ đầu. Cũng giống như trước, các hàm cơ sở được chọn là các hàm chẵn. Và ta có thể chọn chúng là các hàm sau



$$N_1 = (4 - y^2), N_2 = (4 - y^2)x^2, N_3 = (4 - y^2)y^2, N_4 = (4 - y^2)x^2y^2, N_5 = (4 - y^2)x^4, \dots$$

$$N_1 = 4 - y^2, N_2 = (4 - y^2) x^2, N_3 = (4 - y^2) y^2, N_4 = (4 - y^2) x^2 y^2, N_5 = (4 - y^2) x^4, ( ) \quad (2.2.2)$$

..( )

Hàm xấp xỉ được xác định bằng chương trình con

```
>  $\psi := \text{proc}(x, y, M, N) \text{ option operator, arrow; local } m, n;$ 
     $(4 - y^2) * \text{sum}(\text{sum}(C[m, n] * x^{(2 * m)} * y^{(2 * n)}, m = 0 .. M), n = 0 .. N)$ 
    end proc;
```

$$\psi := (x, y, M, N) \rightarrow (4 - y^2) \left( \sum_{n=0}^N \left( \sum_{m=0}^M C_{m, n} x^{2m} y^{2n} \right) \right) \quad (2.2.3)$$

Rõ ràng, hàm này thỏa mãn điều kiện biên tại  $y = \pm 2$  và không thỏa mãn điều kiện biên tại  $x = \pm 3$ . Các điều kiện biên cuối sẽ được điều chỉnh bằng cách nhập chúng vào các phương trình của phương pháp sai lệch trọng số, giống như đã làm với phương trình (3.14). Ta sẽ soạn thảo chương trình con cho phương pháp sai lệch trọng số dưới dạng phương pháp Bubnov-Galerkin.

```
>  $WDM := \text{proc}(M, N) \text{ local } eq, m, n, R, eqns, var, res;$ 
     $R := \text{diff}(\psi(x, y, M, N), x\$2) + \text{diff}(\psi(x, y, M, N), y\$2) + 2 :$ 
     $eqns := \{ \} : var := \{ \} :$ 
    for m from 0 to M do
        for n from 0 to N do
             $eq[m, n] := \text{int}(\text{int}(R * x^{(2 * m)} * y^{(2 * n)}, x = -3 .. 3), y = -2 .. 2) +$ 
             $\text{int}(\psi(3, y, M, N) * (3)^{(2 * m)} * y^{(2 * n)}, y = -2 .. 2) +$ 
             $\text{int}(\psi(-3, y, M, N) * (-3)^{(2 * m)} * y^{(2 * n)}, y = -2 .. 2) = 0 :$ 
             $eqns := eqns \text{ union } \{eq[m, n]\} :$ 
             $var := var \text{ union } \{C[m, n]\} :$ 
        end do:
    end do:
     $res := \text{solve}(eqns, var) : \text{assign}(res) :$ 
    print(res);
    end proc;
```

Bây giờ ta sẽ lập nghiệm xấp xỉ của bài toán. Xấp xỉ hai thành phần cho kết quả

```
>  $M := 1; N := 0; C := \text{array}(0 .. M, 0 .. N); \psi(x, y, M, N);$ 
     $WDM(M, N);$ 
```

$$M := 1$$

$$N := 0$$

$$C := \text{array}(0 .. 1, 0 .. 0, [ ])$$

$$(4 - y^2) (C_{0,0} + C_{1,0} x^2)$$

$$\left\{ C_{0,0} = \frac{189}{215}, C_{1,0} = -\frac{6}{43} \right\} \quad (2.2.4)$$

6 thành phần xấp xỉ cho kết quả

```
> M := 2; N := 1; C := array(0..M, 0..N);  $\psi(x, y, M, N)$ ;
WDM(M, N);
```

$$\begin{aligned} & M := 2 \\ & N := 1 \\ & C := \text{array}(0..2, 0..1, [ ]) \\ & (4 - y^2) (C_{0,0} + C_{1,0}x^2 + C_{2,0}x^4 + C_{0,1}y^2 + C_{1,1}x^2y^2 + C_{2,1}x^4y^2) \\ & \left\{ C_{0,0} = \frac{85732515}{107495464}, C_{0,1} = \frac{7238385}{429981856}, C_{1,0} = -\frac{2719325}{53747732}, C_{1,1} = -\frac{1598135}{214990928}, \right. \\ & \left. C_{2,0} = -\frac{579145}{107495464}, C_{2,1} = \frac{1015525}{429981856} \right\} \end{aligned} \quad (2.2.5)$$

12 thành phần xấp xỉ cho kết quả

```
> M := 3; N := 2; C := array(0..M, 0..N);
WDM(M, N); p || M || N := plot( $\psi(3, y, M, N)$ , y = -2..2, color = black) :
plots[display]({p21, p32});
```

$$\begin{aligned} & M := 3 \\ & N := 2 \\ & C := \text{array}(0..3, 0..2, [ ]) \\ & \left\{ C_{0,0} = \frac{1310885803627868450997}{1627129643161177272464}, C_{0,1} = \frac{81342291070232685675}{6508518572644709089856}, C_{0,2} = \right. \\ & -\frac{3574030090516075575}{6508518572644709089856}, C_{1,0} = -\frac{94837923801747660723}{1627129643161177272464}, C_{1,1} = \\ & -\frac{7471680183639285033}{6508518572644709089856}, C_{1,2} = \frac{108711336196051512}{101695602697573579529}, C_{2,0} = \\ & -\frac{5871619372802719305}{1627129643161177272464}, C_{2,1} = \frac{12341772576974973825}{6508518572644709089856}, C_{2,2} = \\ & -\frac{2924995669160090295}{6508518572644709089856}, C_{3,0} = \frac{53423133364431831}{1627129643161177272464}, C_{3,1} = \\ & \left. -\frac{1809096284963338371}{6508518572644709089856}, C_{3,2} = \frac{199933594045968783}{3254259286322354544928} \right\} \end{aligned}$$

Error, (in plots:-display) expecting plot structures but received: {p21}

```
>
```

### ▼ Cách 3: Điều kiện biên tự nhiên

(Xuất hiện đạo hàm trong điều kiện biên)

Ta có thể chọn các hàm cơ sở sao cho nó không thỏa mãn các điều kiện biên ngay từ đầu

### Bài toán 3.3.

Giải bài toán truyền nhiệt dừng trong vật liệu với hệ số truyền nhiệt có dạng hình vuông  $-1 \leq x \leq 1, -1 \leq y \leq 1$ , . Trên các biên  $y = \pm 1$  được giữ nhiệt độ không. Trong khi đó qua các cạnh  $x = \pm 1$  có dòng nhiệt với vận tốc  $\cos\left(\frac{\pi y}{2}\right)$  trên một đơn vị chiều dài.

Phương trình vi phân truyền nhiệt dừng mô tả quá trình này là

$$\frac{\partial^2 \varphi}{\partial x^2} + \frac{\partial^2 \varphi}{\partial y^2} = 0.$$

Điều kiện biên là

$$\varphi \Big|_{y=\pm 1} = 0, \quad \frac{\partial \varphi}{\partial x} \Big|_{x=\pm 1} = \pm \cos\left(\frac{\pi y}{2}\right).$$

### Phương pháp giải

Phương pháp Ritz cho bài toán trên dẫn tới bài toán tìm cực tiểu của phiếm hàm

$$F(u) = (Au, u) - 2(f, u) \equiv \int_{\Omega} uAu \, d\Omega - 2 \int_{\Omega} uf \, d\Omega$$

Lập tổ hợp tuyến tính của  $n$  hàm cơ sở đầu tiên với các hệ số tự do  $a_m$

$$u_M = \sum_{m=1}^M a_m N_m(x, y)$$

Thế  $u_M$  vào phương trình của  $F(u)$ , ta được hàm số mới của  $F$  của  $n$  biến độc lập  $a_m$

$$F(u_M) = \sum_{j,k=1}^M a_j a_k \left( \int_{\Omega} N_k A N_j \, d\Omega \right) - 2 \sum_{j=1}^M a_j \left( \int_{\Omega} N_j f \, d\Omega \right)$$

Chọn các giá trị  $a_m$  sao cho hàm trên đạt giá trị nhỏ nhất. Điểm cực trị đó đạt được tại những giá trị của biến độc lập mà tại đó các đạo hàm cấp một bằng không:

$$\frac{\partial F(u_M)}{\partial a_i} = 0, \quad i = 1, 2, \dots, M.$$

Kết quả lấy vi phân cho ta hệ phương trình tuyến tính của các ẩn  $a_m$  sau

$$\sum_{k=1}^M a_k \left( \int_{\Omega} N_k A N_j \, d\Omega \right) = \left( \int_{\Omega} N_j f \, d\Omega \right), \quad j = 1, 2, \dots, M.$$

### Chương trình con

```

> restart; PDE := Diff(u(x, y), x$2) + Diff(u(x, y), y$2) = 0;
#Nhập phương trình vi phân
PDE :=  $\frac{\partial^2}{\partial x^2} u(x, y) + \frac{\partial^2}{\partial y^2} u(x, y) = 0$  (2.3.2.1)

> phi := proc(x, y, M, N) option operator, arrow; local i, j;
(1 - y^2) * sum(sum(a[i, j] * x^(2 * i) * y^(2 * j), i = 0 .. M), j = 0 .. N);
end proc; #Lập hàm xấp xỉ theo công thức

```

$$\phi := (x, y, M, N) \rightarrow (1 - y^2) \left( \sum_{j=0}^N \left( \sum_{i=0}^M a_{i,j} x^{2i} y^{2j} \right) \right) \quad (2.3.2.2)$$

```

> SOL := proc(M, N) local F, eqns, var, eq, i, j, res;
  #Chương trình con cho phương pháp Ritz
  F := simplify(
    int(int(Dphi_x(x, y, M, N)^2 + Dphi_y(x, y, M, N)^2,
    y=-1..1), x=-1..1)
    - 2 * int(phi(1, y, M, N) * cos(pi*y/2), y=-1..1)
    - 2 * int(phi(-1, y, M, N) * cos(pi*y/2), y=-1..1));
  eqns := { } : var := { } :
  for i from 0 to M do
    for j from 0 to N do
      var := var union {a[i, j]} :
      eq[i, j] := diff(F, a[i, j]) = 0 :
      eqns := eqns union {eq[i, j]} :
    od;
  od;
  res := solve(eqns, var);
  assign(res);
end proc;

```

**SOL := proc(M, N)** (2.3.2.3)

```

local F, eqns, var, eq, i, j, res;
F := simplify(int(int(Dphi_x(x, y, M, N)^2 + Dphi_y(x, y, M, N)^2, y= - 1
..1), x= - 1..1) - 2 * (int(phi(1, y, M, N) * cos(1/2 * pi*y), y= - 1
..1)) - 2 * (int(phi(-1, y, M, N) * cos(1/2 * pi*y), y= - 1..1));
eqns := { };
var := { };
for i from 0 to M do
  for j from 0 to N do
    var := union(var, {a[i, j]});
    eq[i, j] := diff(F, a[i, j]) = 0;
    eqns := union(eqns, {eq[i, j]})
  end do
end do;
res := solve(eqns, var);
assign(res)
end proc

```

```

> dx := simplify(diff(phi(x, y, M, N), x));
  # `Các đạo hàm riêng (các thành phần của gradient)
dy := simplify(diff(phi(x, y, M, N), y));
Dphi_x := unapply(dx, [x, y, M, N]); Dphi_y := unapply(dy, [x, y, M, N]);

```

$$\begin{aligned}
 dx &:= -2 (-1 + y^2) \left( \sum_{j=0}^N y^{2j} \left( \sum_{i=0}^M x^{2i-1} i a_{i,j} \right) \right) \\
 dy &:= -2 y \left( \sum_{j=0}^N y^{2j} \left( \sum_{i=0}^M a_{i,j} x^{2i} \right) \right) + 2 \left( \sum_{j=0}^N j y^{2j-1} \left( \sum_{i=0}^M a_{i,j} x^{2i} \right) \right) - 2 \left( \sum_{j=0}^N j y^{2j-1} \left( \sum_{i=0}^M a_{i,j} x^{2i} \right) \right) y^2 \\
 Dphi\_x &:= (x, y, M, N) \rightarrow -2 (-1 + y^2) \left( \sum_{j=0}^N y^{2j} \left( \sum_{i=0}^M x^{2i-1} i a_{i,j} \right) \right) \\
 Dphi\_y &:= (x, y, M, N) \rightarrow -2 y \left( \sum_{j=0}^N y^{2j} \left( \sum_{i=0}^M a_{i,j} x^{2i} \right) \right) + 2 \left( \sum_{j=0}^N j y^{2j-1} \left( \sum_{i=0}^M a_{i,j} x^{2i} \right) \right) - 2 \left( \sum_{j=0}^N j y^{2j-1} \left( \sum_{i=0}^M a_{i,j} x^{2i} \right) \right) y^2
 \end{aligned}
 \tag{2.3.2.4}$$

```

> M := 1; N := 0; a := array(0..M, 0..N); i := 'i'; j := 'j';
  phi(x, y, M, N); Dphi_x(x, y, M, N); Dphi_y(x, y, M, N);
  SOL(M, N);
  py_ || M || N :=
  plot(Dphi_x(1, y, M, N), y = 0..1, color = black, linestyle = 3,
  legend = `2 elements of approximation`);
  phi(x, y, M, N);

```

$$\begin{aligned}
 M &:= 1 \\
 N &:= 0 \\
 a &:= \text{array}(0..1, 0..0, [ ]) \\
 &(1 - y^2) (a_{0,0} + a_{1,0} x^2) \\
 &\quad - 2 (-1 + y^2) x a_{1,0} \\
 &\quad - 2 y (a_{0,0} + a_{1,0} x^2) \\
 &(1 - y^2) \left( \frac{54}{7 \pi^3} + \frac{90}{7} \frac{x^2}{\pi^3} \right)
 \end{aligned}
 \tag{2.3.2.5}$$



## Tài liệu tham khảo

- [1.] Dương Thủy Vỹ (2007), Phương pháp tính, NXB Khoa học và kỹ thuật.
- [2.] Nguyễn Hữu Điền (2015), Thực hành tính toán trong Maple, NXB Đại học quốc gia Hà Nội.
- [3.] Nguyễn Thế Hoàn và Phạm Phú (2007), Cơ sở phương trình vi phân và lý thuyết ổn định, NXB Giáo dục.
- [4.] Nguyễn Thế Hoàn và Trần Văn Nhung (2009), Bài tập phương trình vi phân, NXB Giáo dục.
- [5.] Heckbert P. (2000), Partial Differential Equations, Computer Science Department, Carnegie Mellon University.
- [6.] Phan Huy Thiện (2006), Phương trình toán lý, NXB Giáo dục.
- [7.] Phan Văn Cường (nhóm niên luận Toán tin K32 ĐH Khoa học Huế) (2014), Sơ lược về Maplet trong Maple  
<https://huynhquysp.files.wordpress.com/2014/08/so-luoc-ve-maplet-trong-maple.pdf>
- [8.] Richard L. Burden, J. Douglas Faires (2011), Numerical analysis, Youngstown State University.
- [9.] Tạ Văn Đĩnh (1994), Phương pháp tính, NXB Giáo dục.